
INTRODUCTION 
 
This document supplements the PSpice model DM6V6.INC, and provides some 
background operation to the operation of the model along with details of functionality 
modelled or not modelled as the case may be. 
 
Whilst every care has been taken to duplicate the functionality of the modelled device 
as described here, it should be stressed that modelling is not a substitution for 
breadboarding or other prototyping methods. 
 
No warranty of any kind is provided for this model, and no liability is assumed for any 
damage or loss arising out of the use of this model, or application of the results of this model.  
All trademarks acknowledged.  The model is copyright ©1997-2003 Duncan Amplification, 
and is made available for educational or non-profit use. 
 
 
MODELLED FUNCTIONS 
 
Inter-electrode capacitance, screen current, grid current. 
 
 
FUNCTIONS NOT MODELLED 
 
A heater model is not implemented at this stage. 
 
 
MODEL PERFORMANCE 
 

 
Figure 1: Anode current 

 
The diagram above shows anode current against swept anode voltage for a range of 
grid voltages between -30 and +20V in steps of 5V.  Screen voltage in this instance 
is 250V. 

Note how the left-hand side of each trace reaches a saturation point, or knee, 
accurately reflecting the true device behaviour. 
 
The next diagram shows screen current overlaid on the anode current chart.  Grid 
voltage and screen voltages are the same as the previous diagram. 

 
As can be clearly seen, the screen current rises sharply as conduction approaches 
the knee. 
 

 
Figure 3: Anode current against grid voltage 

 

Figure 2: Anode and screen current together 

The chart above shows anode current against grid voltage.  The numbers against 
each plot show the static screen and anode voltage. 
 
 
MODEL DESCRIPTION 
 
The following describes the various components of the model and their interaction: 
 
Eat  is the arctangent calculation which causes the fall off in emission at lower 

anode voltages.  This calculation is normally used for pentodes, although 
for beam tetrodes (such as the 6V6), a sharper knee is applied over the 
top later on. 

 
Eme is the slope, which sets the knee value. 
 
Egs is the emission contribution from the grid and screen, g1 and g2. 
 
Egs2 is Egs after raising to the power of 3/2 and factored by a constant so that 

it may be turned directly into a current value. 
 
Ecath is the cathode current value.  This is the current between anode and 

cathode, although some of this may be diverted by the screen grid.  
Basically consists of Egs2 multiplied by Eat, and then limited by Eme.  
Although this causes a sharp knee cut-off (as evident by the charts), it is 
a "quick and dirty" calculation, which is sufficient for most applications. 

 
Ga is the actual cathode current.  Synonymous with Ecath. 
 
Escrn is the screen current value. 
 
Gs is the actual screen current.  Synonymous with Escrn. 
 
Gg is the grid current value. 
 
 
ALTERING THE MODEL FOR OTHER SIMULATORS 
 
It may be necessary to use the model with other simulators, such as Berkeley SPICE 
3f4, in which case some of the PSpice specific items will need to be altered. 
 
The PSpice LIMIT{a,b,c} statement can, in instances where b is zero, be replaced by 
the SPICE 3f4 statement URAMP(a).  Where LIMIT{a,b,c} is used, with b=0 and 
c=variable, the SPICE 3f4 statement U(a/c)*c can be used. 
 
 
Duncan Munro 
10th May 1997 
 
 
Web page: http://www.duncanamps.com/ 
Forum:  http://forum.duncanamps.com/ 
Email:  postmaster@duncanamps.com 

MODEL LISTING 
 
 
 
PSpice is a trademark of MicroSim Corporation. 
 

* 
* DM6V6.INC - PSpice Subcircuit for 6V6 output beam tetrode 
* 
* Supported:   screen current and interelectrode capacitances, also grid 
*              current with rise at low va. 
* 
* Unsupported: Heater model. 
* 
* D.Munro - 04/05/97 
* 
* 04/05/97  Initial model. 
* 
* 
* Pins   A  Anode 
*        S  Screen 
*        G  Grid 
*        K  Cathode 
* 
.SUBCKT 6V6 A S G K 
* 
* Calculate contribution to cathode current 
* 
Eat at 0 VALUE={0.636*ATAN(V(A,K)/10)} 
Eme me 0 VALUE={PWR(LIMIT{V(A,K),0,2000},1.5)/1300} 
Egs gs 0 VALUE={LIMIT{V(A,K)/600+V(S,K)/14+V(G,K)*0.65,0,1E6}} 
Egs2 gs2 0 VALUE={PWRS(V(gs),1.5)*1.45E-3} 
Ecath   cc     0  VALUE={LIMIT{V(gs2)*V(at),0,V(me)}} 
* 
* Calculate anode current 
* 
Ga      A      K      VALUE={V(cc)} 
* 
* Calculate screen current 
* 
Escrn sc 0 VALUE={0.7*V(gs2)*(1.1-V(at))} 
Gs      S      K VALUE={V(sc)} 
* 
* Grid current 
* 
Gg G K VALUE={PWR(LIMIT{V(G,K)+1,0,1E6},1.5)*(1.25-V(at))*650E-6} 
* 
* Capacitances 
* 
Cg1 G K 7.5p 
Cak A K 9p 
Cg1a G A 0.7p 
 
.ENDS 


